BASIC programming

Subroutine

There are many problems in which a set of calculations or instructions have to be repeated a number of times but with different variable each time. In such cases, the set of statements is written as subprogram with a dummy variable. The subprogram is known as subroutine. It is written at the end of the program after the END statement. Whenever the subroutine statements are required, a value is assigned to dummy variable and the subroutine is accessed by using **GOSUB** statement. The last statement of the subroutine is RETURN. Therefore, as soon as the program reaches RETURN statement, it goes back to the main part of the program. The subroutine is written after the END statement so that the program will not go to the subroutine automatically. The syntax of using a subroutine is,

```
20 P = 5: R= 4
30 X = P
40 GOSUB 1000
50 A = S
60 X = R
70 GOSUB 1000
80 B = S
….
….
….
200 END
1000 FOR I = 1 TO X
……
…….
1040 NEXT I
1050 RETURN
```

Problem 23: Write a program in BASIC to find the binomial coefficient

$$^{n}C_{r} = \frac{n!}{r!(n-r)!}$$

Program: Text in red is explanation and it it not to be written in the main program

```
10 CLS
20 REM* VALUE OF BINOMIAL COEFFICIENT*
30 INPUT "THE VALUE OF n"; N
40 INPUT "THE VALUE OF r"; R
50 P = N    This statement assigns the value of N to dummy variable P
60 GOSUB 500      This statement takes the control to subroutine
70 A = S          This statement saves the value of S which is N! as A
80 P = R          This statement assigns the value of R to dummy variable P
90 GOSUB 500      This statement takes the control to subroutine
100 B = S         This statement saves the value of S which is R! as B
110 P = N - R     This statement assigns the value of N-R to dummy variable P
120 GOSUB 500     This statement takes the control to subroutine
130 C = S         This statement saves the value of S which is (N-R)! as C
```

Dr. Neeti Misra, Associate Professor, Department of Chemistry,
Acharya Narendra Dev College (University of Delhi)
neetimisra@andc.du.ac.in

```
140 ncr = A / (B * C)
150 PRINT "The value of binomial coefficient is "; ncr
160 END
```

Now starts subroutine. The highlighted portion is subroutine. P is the dummy variable.

```
500 S = 1
510 IF P > 34 THEN 560
520 FOR I = 1 TO P
530 S = S * I
540 NEXT I
550 GOTO 570
560 PRINT "THE VALUE IS OUT OF RANGE": END
570 RETURN   This statement takes the control back to the main program
```

Multiple Branching- ON-GOTO statement

The ON-GOTO statement is used to transfer the control of the program to different statement numbers depending on the condition satisfied. Thus, unlike IF-THEN-ELSE statement, that can transfer control to only two statements, ON-GOTO is used for multiple branching. The syntax for using the command is

ON M GOTO N1, N2, N3, N4

M has integral values, 1, 2, 3 and so on. N1, N2, N3, N4 are statement numbers. If the value of M is 1 then the control is transferred to statement N1, if the value of M is 2 the control is transferred to statement N2.

Problem 24: Write a program in BASIC to calculate the pressure of a gas using

(i) ideal gas equation (ii) van der Waal's equation (iii) Dietrici equation.

Program:

```
10 CLS
20 REM*PROGRAM TO CALCULATE PRESSURE OF A GAS*
30 PRINT " ENTER 1 FOR IDEAL GAS EQ, 2-FOR VAN DER WAAL EQ,3-DIETERICI EQ"
40 INPUT M
50 INPUT "Temperature of gas in K "; T
60 INPUT "Volume of gas in m^3"; V
70 R = 8.314
80 INPUT "Moles of gas"; N
90 ON M GOTO 100, 140, 190    If you input 1 in statement 40, control jumps to 100, if
                              input is 2 control jumps to statement 140 and if input is 3 it
                              goes to statement 190
95 REM** Start of ideal gas calculations***
100 PRINT "PRESSURE OF AN IDEAL GAS AT TEMP "; T; "VOL "; V; " IS"
110 P = N * R * T / V
120 PRINT P
130 GOTO 230
135 REM **start of vdW calculations
140 INPUT " ENTER VALUES OF A,B IN SI UNITS"; A, B
150 P = ((N * R * T) / (V - N * B)) - ((A * N ^ 2) / (V ^ 2))
160 PRINT "PRESSURE OF VANDER WAAL GAS AT TEMP"; T;
170 PRINT " AND VOL "; V; " IS "; P
180 GOTO 230
```


Dr. Neeti Misra, Associate Professor, Department of Chemistry,
Acharya Narendra Dev College (University of Delhi)
neetimisra@andc.du.ac.in

```
185 REM*** Calculations from Dieterici equation***
190 INPUT "DIETERICI CONSTANTS A,B IN SI UNITS"; A, B
200 P = ((N * R * T) / (V - B)) * EXP(A / (R * T * V))
210 PRINT "PRESSURE OF DIETRICI GAS AT TEMP"; T;
220 PRINT " AND VOL "; V; " IS "; P
230 PRINT "IF YOU WANT TO CONTINUE ENTER Y ELSE N"
240 INPUT Y$       Enter Y if you want to continue
250 IF Y$ = "Y" OR "y" THEN 30 If the value of Y$ is Y or y then control goes back to
                               30
260 END
```

Output:

```
ENTER 1 FOR IDEAL GAS EQ, 2-FOR VAN DER WAAL EQ,3-DIETERICI EQ
? 1
Temperature of gas ? 300
Volume of gas in m^3? 22.414E-3
Moles of gas? 1
PRESSURE OF IDEAL GAS AT TEMP  300 VOL  .022414  IS
 111278.7
IF YOU WANT TO CONTINUE ENTER Y ELSE N
? Y
 ENTER 1 FOR IDEAL GAS EQ, 2-FOR VAN DER WAAL EQ,3-DIETERICI EQ
? 2
Temperature of gas ? 300
Volume of gas in m^3? 22.414E-3
Moles of gas? 1
 ENTER VALUES OF A,B IN SI UNITS? 21.76E-6,.02661E-3
PRESSURE OF VANDER WAAL GAS AT TEMP 300  AND VOL  .022414  IS  111410.9
IF YOU WANT TO CONTINUE ENTER Y ELSE N
? N
```

Program: Write a program in BASIC to calculate the relative intensities of peaks of a proton obtained after spin-spin coupling with 4 equivalent neighbouring protons in a high resolution NMR spectrum.

Chemistry: Magnetic fields of adjacent ^{1}H can interact with each other. This is known as spin-spin coupling. The number of peaks into which a signal splits due to spin-spin coupling is equal to n+1 where n is the number of neighbouring protons.
The relative intensities of the peaks split due to spin-spin coupling can be predicted by the Pascal triangle or by the relation

$$^nC_m = \frac{n!}{m!(n-m)!}$$

Here, n = number of neighbouring protons, m goes from 0 to n

For n = 4 number of peaks = n+1 = 5

For first peak m = 0, ^nC_m = $^5C_0$ = 1

For second peak m = 1, $^5C_1$ = 1 and so on

```
10 CLS
20 REM RELATIVE INTENSITY OF PEAKS IN SPIN-SPIN COUPLING IN NMR
30 INPUT "NO. OF NEIGHBOURING PROTONS=", N
40 X = N: GOSUB 500
50 NFAC = P
55 PRINT "NUMBER OF PEAKS INTO WHICH THE SIGNAL SPLITS="; N + 1
60 PRINT "RATIO OF INTENSITIES OF THE PEAK IS"
70 FOR M = 0 TO N
80 I = M + 1
90 X = M: GOSUB 500
100 MFAC = P
110 X = N - M: GOSUB 500
120 C(I) = NFAC / (MFAC * P)
130 PRINT C(I);
140 IF M = N THEN 160 ELSE 150
150 PRINT ":";
160 NEXT M
170 END
500 P = 1
510 FOR J = 1 TO X
520 P = P * J
530 NEXT J
540 RETURN
```