MEAN AND STANDARD DEVIATION

Problem: Write a program in BASIC to find the mean and standard deviation for the given data. 1.99,1.87,1.925,1.895,1.989

Mean = $\bar{x} = \frac{x_1 + x_2 + x_3 + \dots\dots\dots + x_N}{N}$ and Standard deviation = $\sqrt{\frac{\sum (xi - \bar{x})^2}{N}}$

$$\bar{x} = \frac{\sum_{i=1}^{N} x_i}{N}$$

x_1, x_2......are the values of x
N= number of values.
Standard deviation is $(\text{variance})^{1/2}$
For calculation of mean, sum of all N *x* values are required.
In the program, this sum $\sum_{i=1}^{N} x_i$ is represented by S. For calculation of variance, again we have to sum the square of difference of each value from mean value, that is, $\sum (xi - \bar{x})^2$. This sum is represented by the variable S1. Since N values of x_i are required, we use 1-D arrays.

Program:

```
10 CLS
20 REM* MEAN AND STANDARD DEVIATION*
30 INPUT "No. of variables="; N      Enter the number of data points
40 DIM X(N)              Reserve N slots for 1-D array X(I)
50 S = 0: S1 = 0         Initializing the variables
60 FOR I = 1 TO N        First value of I is 1, then 2…. last value is N.
                         So the next two steps will be repeated N times
70 READ X(I)             Read each data point and assign the variable X(I)
80 S = S + X(I)          Initial value of S is 0, in first step, the value
                         of X(1) is added to S, in next step value of X(2)
                         is added and so on.
100 NEXT I               The value of I increases by 1 in each step till it
                         is N. After that the control comes out of the loop.
110 M = S / N            Mean value calculation.
120 FOR I = 1 TO N       Loop started for the calculation of variance.
130 S1 = S1 + (X(I) - M) ^ 2    BASIC form of ∑(xi−x̄)²
140 NEXT I
150 VAR = S1 / N         Calculation of variance
160 SD = SQR(VAR)        Calculation of standard deviation
170 PRINT "MEAN="; M
180 PRINT "STANDARD DEVIATION="; SD
190 DATA 1.99,1.87,1.925,1.895,1.989
200 END
```

PROGRAM FOR QUITEBASIC.COM

```
10 CLS
20 REM* MEAN AND STANDARD DEVIATION*
30 INPUT "No. of variables="; N
40 DIM X(N)
50 LET S = 0: LET S1 = 0
```

```
60 FOR I = 1 TO N
70 READ X(I)
80 LET S = (S + X(I))
100 NEXT I
110 LET M = (S / N)
120 FOR I = 1 TO N
130 LET S1 = (S1 + ((X(I) - M)* (X(I) - M)))
140 NEXT I
150 LET V1 = (S1/N)
160 LET D = SQR(V1)
170 PRINT "MEAN="; M
180 PRINT "STANDARD DEVIATION="; D
190 DATA 1.99,1.87,1.925,1.895,1.989
200 END
```

Solution of polynomial equation: (Part 1)

We have seen the method for solving quadratic equation. Now we are going to see how a solution for any polynomial equation can be obtained. The polynomial is of the type.

$f(x) = ax^n + bx^{n-1} + \ldots\ldots.. x + k$ …..(1)

Solution means, the value of x for which f(x) = 0.

In chemistry, we come across many polynomial equations. These are generally simplified by applying assumptions. However, these simplified forms of the equations give approximate results. To get better results, the polynomial equation has to be solved in the exact form.

There are a large number of mathematical methods for solving polynomial equations. We are going to learn a few methods in this paper.

Iterative method

Iterative means something that is repeated a number of times. So Iterative method means a problem solving process in which the steps are repeated more than once before the result is obtained.

In this method the polynomial equation is written as

$x = ax^n + bx^{n-1} + \ldots\ldots.. + k$ …..(2)

a,b,….., k are constants with known values.

An approximate value of x (known as initial guess) is put in the right hand side of the equation and a new value of x (which is taken as x_1) is calculated. x_1 and x are compared. If the difference is within the acceptable limit, one of the solution of equation is x_1. If the difference is large, then x_1 is substituted in the right hand side of the equation and second value x_2 is calculated. This procedure is repeated until the difference between two successive values becomes less than the accepted limit. If the difference increases in successive iterations, then the initial choice of x is incorrect and a new initial value of x has to be taken.

In chemistry, for most of the equations, the initial guess is generally the value of x calculated from the approximate form of the polynomial equation.

The following program illustrates this.

Problem: Write a program in BASIC to solve numerically van der Waal's equation of state using iterative method.

$$\left(P+\frac{an^2}{V^2}\right)(V-nb)=nRT \qquad \ldots.(3)$$

Logic: van der Waal's equation of state is a cubic equation. To solve it using iterative method, the equation is written in the required form.

$$V = nRT/\left(P+\frac{an^2}{V^2}\right) + nb \qquad \ldots\ldots.(4)$$

The initial value of V is the value of volume calculated using ideal gas equation $pV = nRT$. Put this value in the RHS of the eq(4). The values of rest of the variables in the equation are known. The LHS V is taken as $V1$, to make it different from initial value, so equation 4 becomes,

$$V1 = nRT/\left(P+\frac{an^2}{V^2}\right) + nb \qquad \ldots.(5)$$

The difference between, $V1$ and V is calculated. Since, this difference can be positive or negative, the value is taken without sign. The difference is divided to get the relative difference.

$|V1 - V|/V$ V=0.01 V1=0.011 V1-V = 0.001 (0.001/0.1)x100 = 10%

Just to give you an idea, if you multiply the relative difference you get the percentage of the difference in the value. For more accurate result, this relative difference should be small. If the difference is larger than the required accuracy then the procedure is repeated, but now the value of $V1$ is put in the RHS of the equation and a new value of V is calculated.

Program: Take P = 1 atm, T = 273 K, moles = 1 and number of iterations =50

```
10 CLS
20 REM*ITERATIVE METHOD*
30 INPUT "PRESSURE IN atm"; P
40 INPUT "TEMPERATURE IN Kelvin"; T
50 INPUT "MOLES"; N
60 INPUT "NO. OF ITERATIONS"; M
70 A = 1.5: B = .02: R = .082        Values are not in SI
80 V = N * R * T / P                 Calculation of initial value
90 PRINT "INITIAL VALUE OF V="; V
100 FOR I = 1 TO M             Starting the iterations
110 V1 = N * R * T / (P + (A * N ^ 2) / (V ^ 2)) + N*B  Writing Eq (5) in
                                    BASIC
```


Dr. Neeti Misra ,Associate Professor, Department of Chemistry
Acharya Narendra Dev College(University of Delhi)
neetimisra@andc.du.ac.in

```
120 IF ABS((V1 - V) / V) < .0001 THEN 170   Calculation of relative
                    difference, .0001 is accuracy required
130 V = V1          Starting 2nd iteration by taking the value of V1 as new V
140 NEXT I
150 PRINT "THE VALUES HAVE NOT CONVERGED"  You will reach here only when
                  the relative difference does not become less than .0001
                  even after repeating the process 50 times.
160 GOTO 180
170 PRINT "THE VOLUME OF GAS IS "; V1   You will reach here when the
                  relative difference become less than .0001
180 END
If you are getting the meassage THE VALUES HAVE NOT CONVERGED, try
decreasing the accuracy value to .001 and run the program again.
```

(b) Newton-Raphson method

In this method the polynomial equation is written in the form f(x) = 0.

The recursive formula is applied.

$$x_{k+1} = x_k - \frac{f(x_k)}{f'(x_k)} \qquad .....(6)$$

$f(x_k)$ is the value of function when $x = x_k$. Here, x_k is the initial guess, that is the value near the expected solution value. Put the value of x_k in the equation, if it is the actual solution, the value will come out to be zero, otherwise, it will have a non-zero value.

$f'(x_k)$ is the first derivative of the function at x_k. Iteration is carried out, that is the values of x_k, $f(x_k)$ and $f'(x_k)$ are put in RHS of eq(6). x_{k+1} is calculated and x_{k+1} is compared with x_k. If the two values are not sufficiently close then the value of x_{k+1} is substituted into the right hand side of the recursion equation (Eq. 6) and the computation is repeated.

Problem: Write a program in BASIC to solve numerically the exact expression for acid dissociation to calculate the pH of a weak acid.

$pH = -\log_{10} [H^+]$

where the exact equation to calculate $[H^+]$ is

$$K_a = \frac{[H^+]\left\{[H^+] - \frac{K_W}{[H^+]}\right\}}{C_a - [H^+] + \frac{K_W}{[H^+]}} \qquad ....(7)$$

Its form as f(x) = 0 is here $x = [H^+]$

$[H^+]^3 + K_a[H^+]^2 - (K_w + K_aC_a)[H^+] - K_aK_w = 0$ ...(8) f(x)

Differential form of eq(8) is

$3[H^+]^2 + 2K_a[H^+] - (K_w + K_aC_a)$ ....(9) f '(x)

$K_w = 1x10^{-14}$ at 25 °C

K_a is acid dissociation constant of acid, C_a is the initial concentration of acid. For weak acid the concentration of H^+ ions is calculated using the following equation

$$[H^+] = \sqrt{K_a C_a} \qquad ....(10)$$

In this program we using the following library functions: DEF FN to define functions, SQR and LOG

LOG(x) = ln x

Program:

```
10 CLS
20 REM*pH OF WEAK ACID
30 KW = 1E-14
40 INPUT "DISSOCIATION CONSTANT OF ACID "; KA
50 INPUT "INITIAL CONCENTRATION OF ACID"; CA
60 DEF FNA (H) = H ^ 3 + KA * H ^ 2 - (KW + KA * CA) * H - KA * KW  Here
                        LHS is DEF FNA(H)  where A is function name and H is
                        dummy variable used in RHS of the equation. RHS is
                        eq(8)in BASIC syntax.
70 DEF FND (H) = 3 * H ^ 2 + 2 * KA * H - (KW + KA * CA) Here LHS is DEF
                        FND(H)  where D is function name and H is dummy variable
                        used in RHS of the equation. RHS is eq(9)in BASIC
                        syntax. It is derivative

80 H = SQR(KA * CA) Calculation of approximate value(initial value) of H+
                        using Eq(10)
90 FOR I = 1 TO 100     Iterations begin
100 H0 = H        H0 is the variable used for initial value
110  Y  =  FNA(H0):  D  =  FND(H0)    Calculation of value of function and
                        derivative of function at H0
120 H = H0 - Y / D  Recursive formula, Eq (6) applied
130 IF ABS((H - H0) / H0) < .00001 THEN 150          Comparing    relative
                        difference with the required accuracy.
140 NEXT I
150 PRINT "CONC. OF H+ IS "; H
160 PH = -LOG(H) / LOG(10)     This gives the minus of log10 value of H
170 PRINT "pH OF ACID IS "; PH
180 END
```

LOG(10) = ln(10) = 2.303
lnx =2.303 log x

(c) Binary bisection method

This method is based on the repeated application of the intermediate value theorem. The polynomial equation is written in the form f(x) = 0. If the root of the equation lies in the

interval [a, b] then the sign of the function at the two extremes of the interval is different, that is,
sign of f(a) ≠ sign of f(b)



It is assumed that the root of the equation lies at the midpoint of the interval [a, b]. The mid-point c is calculated as

$$c = \frac{a+b}{2}$$

Calculate the value of function at c, that is , f(c). If f(c) = 0 then c is the required root. Otherwise check the subintervals [a,c] and [c,b]. If the sign of f(c) is same as f(a) then the root does not lie in the interval [a,c]. It lies in the interval [b,c]. Otherwise the root lies in the interval [a,c]. Now find the midpoint of the subinterval in which the root lies. If the interval is [a,c] then the midpoint is

$$c' = \frac{a+c}{2}$$

If the interval is [c,b] then the midpoint is

$$c' = \frac{c+b}{2}$$

Calculate f(c´). Repeat the process until the exact root is obtained. If the interval [a,b] is large then the method is modified. The entire interval is subdivided into small intervals of range s. For the first step the interval is a and a+s. This interval is checked for the presence of the root. If the root is absent then next interval between a+s and a+2s is taken. This process is continued until either the root is obtained or maximum value of interval, b is reached.

Problem: Solve the equation $x^5 + 2x^4 + 4x = 5$ using binary bisection method for the root that lies between 0 and 1.

Logic: In this program, lower limit is x_1 and upper limit is x_{max}. So the interval [a, b] is [x_1, x_{max}]. However, the entire interval range is divided into smaller range and we atart checking from x_1 in the intervals of S. So first we check if the root lies between x_1 and x_1+S. If f(x_1) = 0 then the root is x_1. If f(x1+S) =0 then the root is x1+ S. If the sign of the function at x_1, that is f(x_1) and the sign of the function at x_1+S, that is f(x_1+S) are same that means the root does not lie in the range. If the signs are different then the root lies in the range. If root does not lie between x_1 and x_1+S, then we check between x_1+S and x_1+2S and so on until the root is obtained or upper limit x_{max} is reached. If upper limit is reached then that means there are no roots in the range.

Program:

```
10 CLS
20 REM*BINARY BISECTION METHOD
30 INPUT "LOWER LIMIT"; X1
40 INPUT "UPPER LIMIT"; XMAX
50 INPUT "STEP"; S
60 DEF FNA (X) = X ^ 5 + 2 * X ^ 4 + 4 * X - 5  polynomial is written
70 X2 = X1 + S
80 IF X2 > XMAX THEN PRINT "NO ROOTS IN TNE RANGE": END
90 Y1 = FNA(X1): Y2 = FNA(X2)  Value of f(x1) and f(x1+S) is calculated.
100 IF Y1 = 0 THEN PRINT "ROOT="; X1: END
110 IF Y2 = 0 THEN PRINT "ROOT="; X2: END
120 IF SGN(Y1) = SGN(Y2) THEN 130 ELSE 150      Comparing the signs of f(x1)
                                            and f(x1+S)
130 X1 = X2       We reach here when the root does not lie between x1 and
                  x1+S, so we move to the next range and take the new lower
                  limit as x2 which is x1+S
140 GOTO 70       At 70, a new upper limit x1+2S is calculated
150 X3 = (X1 + X2) / 2     We reach here when the root lies between x1 &
                  x2. Mid-point of the range is calculated
160 Y3 = FNA(X3)  The value of the function at midpoint.
170 IF Y3 = 0 THEN 220
180 IF ABS((X2 - X1) / X3) <= .00001 THEN 220   Checking if the range is
                       very small, so x1 & x2 are very close to each other.
190 IF SGN(Y3) = SGN(Y1) THEN 200 ELSE 210
200 X1 = X3: Y1 = Y3: GOTO 150
210 X2 = X3: Y2 = Y3: GOTO 150
220 PRINT "ROOT="; X3
230 END
```

Output:

```
LOWER LIMIT? 0
UPPER LIMIT? 1
STEP? .1
ROOT= ………………..
```

Problem 30: The van der Waals equation for argon at 143 K and 35.0 atm pressure can be written as

$$\overline{V}^3 - 0.3664\overline{V}^2 + 0.03802\overline{V} - 0.001210 = 0$$

Write a program in BASIC to find the roots to this equation using Newton-Raphson method.